


Tutorial: building a multi-bank ROM image

Introduction

This tutorial describes the steps involved for building a multi-bank cartridge ROM image for the Texas Instruments TI-99/4A Home Computer.

As an example we'll build a Pitfall! cartridge ROM image; starting from the assembly source. When done, we will have a 32K ROM image containing four 8K (>6000->7FFF) banks. Burned on EPROM it can be used with the new 16K/32K/64K cartridge PCBs^[1] recently made available by Jon Guidry.

Our target platform: the 16K/32K/64K cartridge PCB



© Picture courtesy of Jon Guidry

Note that all steps described in the tutorial will be done on a windows based cross-development system. The main reason is, that the cross-assembler I use is only available for Windows.

[1] <http://www.hexbus.com>

Requirements

You will need a recent version of the windows based cross-assembler **winasm99** which is delivered as part of the win994A emulator package^[2]. I used winasm99 v3.8

Of course you'll also need the Pitfall! source code^[3] and my deref command-line utility^[4]. The latter is mainly used for handling code-realignment issues. More on that later.

[2] <http://www.99er.net/win994a.shtml>

[3] <http://www.atariage.com/forums/topic/158467-pitfall-binaries-and-source-code/>

[4] <http://www.atariage.com/forums/topic/153704-ti-994a-development-resources/>

A few hints

Pre-processor directives

I found out that winasm99 supports pre-processor directives, a feature which I believe is not officially documented (nor supported?). Basically I created a main file for each bank and added the equates below. Depending on the bank I then set the corresponding equate to 1.

```
*****
* Some build directives for building the 2nd bank (BANK1)
*****@*****@*****@*****@*****
BANK0 EQU 0
BANK1 EQU 1
BANK2 EQU 0
BANK3 EQU 0
```

In the next example the source code between 'IF BANK1' and 'ENDIF' is only assembled if the BANK1 equate equals 1. So I can include (COPY) the same source file in multiple projects/main files and control how it gets assembled by setting the corresponding BANK equates to 1 or 0.

```
DRAWBJ LI R1,OVLAY4
        BL @SPRITE ; Put overlay sprite "yellow ground" on screen
        BL @GVRAM
        DATA VRETRN,RETADR,2 ; Restore Return address ...
        MOV @RETADR,R0
* .....
IF BANK1
    CI R0,DRAWG1 ; Called from DRAWG ?
    JEQ DRAWBZ ; Yes, then Return to DRAWG1
    LI R1,>6002 ; Select BANK2
    LI R2,PFRETN ; Jump-back to PFALL in BANK2
    B @GOBANK ; Switch bank 2
ENDIF
* .....
DRAWBZ B *R0 ; ... and Return
```

Bank-switch trampoline code

What you see in the example, is a bank-switch from BANK 1 (2nd bank) to BANK 2 (3rd bank). This is accomplished by calling the GOBANK subroutine which in turn writes the bank-switch code to scratch-pad memory and executes it.

```
*-----
* ROM BANK SELECTION ROUTINE (inversed method for 379 carts)
*-----
* R1=>6006,>6004,>6002,>6000 ; BANK0, BANK1, BANK2, BANK3
* R2=Address in bank
*-----
SWBANK EQU >8300 ; Somewhere in scratchpad RAM
GOBANK LI R0,>04E0 ; CLR ...
        MOV R0,@SWBANK
        MOV R1,@SWBANK+2 ; ... CLR R1 -> Select BANK0 ... BANK3
        LI R0,>0460 ; B ...
        MOV R0,@SWBANK+4
        MOV R2,@SWBANK+6 ; ... B (address in R2)
        B @SWBANK ; Now run generated code
```

BANK 0 >6000->7FFF	BANK 1 >6000->7FFF	BANK 2 >6000->7FFF	BANK 3 >6000->7FFF
Cartridge Header	Cartridge Header	Cartridge Header	Cartridge Header
write >6006	write >6004	write >6002	write >6000
ROMC.bin ROMC.lst	ROMD.bin ROMD.lst	ROME.bin ROME.lst	ROMF.bin ROMF.lst

Cartridge header

Note that the cartridge header must be present in each bank, as the TI-99/4A can fire up a random bank on reset. In each bank the cartridge header points to a routine that switches to BANK 0 and triggers the initialisation routine. Below is a more complete example, it's part of the main file for BANK 1 (2nd bank):

```

AORG >6000
SFIRST EQU $
SLOAD EQU $
*****
* Some build directives
*****@*****@*****@*****@*****@*****@*****
BANK0 EQU 0
BANK1 EQU 1
BANK2 EQU 0
BANK3 EQU 0
*****
* Some equates with addresses of objects in other banks
*****@*****@*****@*****@*****@*****@*****
COPY "D:\Projekte\pitfall\tms9900\pitfall_bank_equates.a99"
*****
* Cartridge header
*****@*****@*****@*****@*****@*****@*****
GRMHDR BYTE >AA,1,1,0,0,0
DATA PROG
BYTE 0,0,0,0,0,0,0,0
PROG DATA 0
DATA KICKBK
BYTE 8
TEXT 'PITFALL!'
*****
* PITFALL BANK 1
*****@*****@*****@*****@*****@*****@*****
COPY "D:\Projekte\pitfall\tms9900\pitfall.a99"
*****
* Kickstart BANK0
*****@*****@*****@*****@*****@*****@*****
KICKBK LWPI WSSPC1 ; Load main workspace
LIMI 0
LI R1,>6006 ; Select BANK0
LI R2,B0MAIN ; B @MAIN (skip cartridge header)
B @GOBANK ; Switch bank 0
*****
* Main program
*****@*****@*****@*****@*****@*****@*****
...
*-----
* Include required files
*-----
COPY "D:\Projekte\pitfall\tms9900\pitfall1a.a99"
COPY "D:\Projekte\pitfall\tms9900\pitfall1c.a99"
...

```

Notice the inclusion of **pitfall_bank_equates.a99** ? That is really important, more on that in the next topic.

Dependencies across banks

The biggest problem one faces during the implementation of a multi-bank ROM image, is having to deal with broken dependencies because of code and data realignment issues.

Suppose you are in subroutine SUBC in BANK 2 and want to jump to subroutine SUBB in BANK 1. Normally the steps you would do are:

- ➔ Get the address of SUBB in BANK 1.
- ➔ Set this address in subroutine SUBC in BANK 2.
- ➔ Assemble BANK 2

So far so good. Your program runs ok and your bank-switching code works just fine. You are a happy camper.

Then a few days later you need to fix a small bug in subroutine SUBA in BANK 1. You just had to add an "AI" statement and reassemble BANK 1. All seems to be ok at first. But then your program crashes after running subroutine SUBC in BANK 2. The problem is that your code still jumps to >6030 in BANK 1 but subroutine SUBB is not longer located there, it is now at >6034.


BANK 1 (BEFORE)		BANK 1 (AFTER)		BANK 2	
>6014	SUBA	>6014	SUBA	>6014	SUBC
	CLR R0		CLR R0		LI R1,>6002
	MOV R1,R2		MOV R1,R2		LI R2,>6030
	RT		AI R1,5		B @GOBANK
>6030	SUBB		RT		...
	CLR R0	>6034	SUBB		
	MOV R1,R2		CLR R0		
	...		MOV R1,R2		
			...		

To resolve the problem BANK 2 needs to be reassembled after adjusting the address in SUBC. Obviously these type of errors can easily result in a debugging nightmare. Updating all the addresses manually is almost an impossible task.

For handling this type of problem I wrote the command-line utility **deref.exe**

It basically scans the list files that get generated by winasm99 during the assembly process. Next it processes the specified template file, replacing the special tags with the corresponding ROM addresses and generate a matching assembly source file.

The command-line utility **deref.exe**



```
C:\WINDOWS\system32\cmd.exe
D:\Projekte\cartridge tools>deref.exe
Usage: deref.exe [OPTION]...

winasm99 post-processor v0.1
<c>2010 by retroclouds <http://www.retroclouds.de>

--sym!file!f=filename      Symbol template file to process

    The template file is searched in the current directory
    if no path is given. The file extension ".sym" will be
    assumed if no extension is provided.

    The script will automatically create a file with extension ".a99"
    having all bank references replaced by the corresponding addresses.

--help!h                  Display this help and exit

--bank0!b0=value          Listing file for bank 0
--bank1!b1=value          Listing file for bank 1
...
--bank15!b15=value        Listing file for bank 15

    The listing file is searched in the current directory
    if no path is given. The file extension ".lst" will be
    assumed if no extension is provided.

EXAMPLES
deref.pl -b0=ROMC -b1=ROMD -b2=ROME -b3=ROMF -f=bank_template
deref.pl -b0=ROMC.lst -b1=ROMD.lst -f=abc.sym -f=def.sym -f=abc2.sym

FORMAT for bank reference tags in template files
<B<0-15>!label>

EXAMPLES
ZLOGO EQU <B0!ZLOGO> ; Address of ZLOGO in BANK0
DRAWB EQU <B1!DRAWB> ; Address of DRAWB in BANK1


D:\Projekte\cartridge tools>
```

The idea is that you manually run deref.exe command-line utility after winasm99 has finished building a bank. It will generate a new assembly source with the updated ROM addresses. You can then run winasm99 for building the next bank.

This is a repetitive process that -depending on the source code modifications- needs to be done multiple times until all pieces fit together.

A template file is basically a plain text file. The only difference compared to normal assembly source files, are the special tags in the format `<B?|label>` where ? is the value of the bank number starting with 0.

The template file "pitfall_bank_equates.sym" used for building the Pitfall! ROM image.



```

C:\WINDOWS\system32\cmd.exe
D:\Projekte\cartridge tools>type pitfall_bank_equates.sym
*.....
IF BANK0
B1GOGA EQU <B1!GOGAME> ; Address of GOGAME in BANK1
B3MAIN EQU <B3!MAIN> ; Address of MAIN in BANK3
ENDIF
*.....
IF BANK1
B0TMGR EQU <B0!TMGR> ; Address of TMGR in BANK0
B0UKBS EQU <B0!UKBSND> ; Address of UKBSND in BANK0
ZPLAY EQU <B0!ZPLAY> ; Address of ZPLAY in BANK0
ZLOGO EQU <B0!ZLOGO> ; Address of ZLOGO in BANK0
ZRLOG EQU <B2!ZRLOG> ; Address of ZRLOG in BANK2
ZSCORP EQU <B2!ZSCORP> ; Address of ZSCORP in BANK2
ZCROCK EQU <B2!ZCROCK> ; Address of ZCROCK in BANK2
ZWATR1 EQU <B2!ZWATR1> ; Address of ZWATR1 in BANK2
ZFIRE EQU <B2!ZFIRE> ; Address of ZFIRE in BANK2
ZROPE EQU <B2!ZROPE> ; Address of ZROPE in BANK2
ZTIMER EQU <B2!ZTIMER> ; Address of ZTIMER in BANK2
HMOVE EQU <B2!HMOVE> ; Address of HMOVE in BANK2
PFALL EQU <B2!PFALL> ; Address of PFALL in BANK2
B0MAIN EQU <B0!MAIN> ; Address of MAIN in BANK0
B2MAIN EQU <B2!MAIN> ; Address of MAIN in BANK2
B2TMGR EQU <B2!TMGR> ; Address of TMGR in BANK2
B2UKBS EQU <B2!UKBSND> ; Address of UKBSND in BANK2
PFRETN EQU <B2!PFRETN> ; Address of PFRETN in BANK2
ENDIF
*.....
IF BANK2
B1OVER EQU <B1!GAOVER> ; Address of GAOVER in BANK1
DRAWB EQU <B1!DRAWB> ; Address of DRAWB in BANK1
B1PAUS EQU <B1!IPAUSE> ; Address of IPAUSE in BANK1
LOG1 EQU <B1!LOG1> ; Address of LOG1 in BANK1
ENDIF
*.....
IF BANK3
B1MAIN EQU <B1!MAIN> ; Address of MAIN in BANK1
ENDIF
*.....
D:\Projekte\cartridge tools>

```

The resulting "pitfall_bank_equates.a99" output file after running the deref.exe command-line utility.

```

C:\WINDOWS\system32\cmd.exe
* Generated by "deref" v0.1 using template pitfall_bank_equates.sym
* Thu Mar 11 20:35:01 2010
*
IF BANK0
B1GOGA EQU >60D2 ; Address of GOGAME in BANK1
B3MAIN EQU >6032 ; Address of MAIN in BANK3
ENDIF
*
IF BANK1
B0TMGR EQU >7D68 ; Address of TMGR in BANK0
B0UKBS EQU >702C ; Address of UKBSND in BANK0
ZPLAY EQU >6338 ; Address of ZPLAY in BANK0
ZLOGO EQU >62FC ; Address of ZLOGO in BANK0
ZRLOG EQU >61DC ; Address of ZRLOG in BANK2
ZSCORP EQU >637A ; Address of ZSCORP in BANK2
ZCROCK EQU >64F4 ; Address of ZCROCK in BANK2
ZWATR1 EQU >645E ; Address of ZWATR1 in BANK2
ZFIRE EQU >64CA ; Address of ZFIRE in BANK2
ZROPE EQU >72F4 ; Address of ZROPE in BANK2
ZTIMER EQU >74DA ; Address of ZTIMER in BANK2
HMOVE EQU >6DDA ; Address of HMOVE in BANK2
PFALL EQU >69DE ; Address of PFALL in BANK2
B0MAIN EQU >601E ; Address of MAIN in BANK0
B2MAIN EQU >6032 ; Address of MAIN in BANK2
B2TMGR EQU >7E32 ; Address of TMGR in BANK2
B2UKBS EQU >7480 ; Address of UKBSND in BANK2
PFRETN EQU >6A9C ; Address of PFRETN in BANK2
ENDIF
*
IF BANK2
B1OVER EQU >606A ; Address of GAOVER in BANK1
DRAWB EQU >61AE ; Address of DRAWB in BANK1
B1PAUS EQU >6BA6 ; Address of IPAUSE in BANK1
LOG1 EQU >739C ; Address of LOG1 in BANK1
ENDIF
*
IF BANK3
B1MAIN EQU >6032 ; Address of MAIN in BANK1
ENDIF
*
D:\Projekte\cartridge tools>

```


Building the multi-bank ROM image

Now let's get on with the fun part; building the multi-bank ROM image. Assuming you have already unzipped the Pitfall! source directory to a local drive/device, you have to do the below steps for successfully creating the cartridge version.

Step 1: Open the files "pitfall_bank0.a99", "pitfall_bank1.a99", "pitfall_bank2.a99", "pitfall_bank3.a99" into your text editor and for each file replace the paths in all COPY statements so that it matches with your local source directory.

Your editors' search & replace functionality will be of great use.

Step 2: After you have saved your modifications, start winasm99 and create a new project file with the settings seen in the next screenshot and click "Start assembly". We are building BANK 0.



Step 3: Check for assembly errors. If all is fine you can then run the deref.exe command-line utility with the below parameters:

deref.exe -b0=ROMC -b1=ROMD -b2=ROME -b3=ROMF -f=pitfall_bank_equates.sym

A new "pitfall_bank_equates.a99" will be generated.

Step 4: Repeat the steps 2 & 3 for the other banks. The best way to do this is to run 4 parallel sessions of winasm99, one for each bank.

Below are the required settings for winasm99

BANK	Input Project	Listing File	Cart binary file
BANK 0	pitfall_bank0.a99	ROMC.lst	ROMC.bin
BANK 1	pitfall_bank1.a99	ROMD.lst	ROMD.bin
BANK 2	pitfall_bank2.a99	ROME.lst	ROME.bin
BANK 3	pitfall_bank3.a99	ROMF.lst	ROMF.bin

Important! When modifying the source code you'll have to repeat the cycle of building banks 0-3 at least 2 times in a row until everything falls into place.

Best thing to do is monitor "pitfall_bank_equates.a99" after each run of the deref.pl script until there are no more changes from version to version.

Step 5: We are almost done. Now concatenate the 4 binary files ROMC.bin, ROMD.bin, ROME.bin, ROMF.bin for getting a single binary file PITFALL.bin

This is real easy when using the windows command window just type the below command:

Copy /B ROMC.bin+ROMD.bin+ROME.bin+ROMF.bin PITFALL.bin

```

C:\WINDOWS\system32\cmd.exe
71 Datei(en)      2.601.612 Bytes
3 Verzeichnis(se), 6.584.238.080 Bytes frei

D:\Projekte\pitfall\tms9900>dir *.bin
Volume in Laufwerk D: hat keine Bezeichnung.
Volumeseriennummer: 68D4-F9FC

Verzeichnis von D:\Projekte\pitfall\tms9900

20.03.2010  20:21             8.192 ROMC.bin
20.03.2010  20:22             8.192 ROMD.bin
20.03.2010  20:22             8.192 ROME.bin
20.03.2010  20:23             8.192 ROMF.bin
4 Datei(en)      32.768 Bytes
0 Verzeichnis(se), 6.584.238.080 Bytes frei

D:\Projekte\pitfall\tms9900>copy /B ROMC.bin+ROMD.bin+ROME.bin+ROMF.bin PITFALL.
bin
ROMC.bin
ROMD.bin
ROME.bin
ROMF.bin
1 Datei(en) kopiert.

D:\Projekte\pitfall\tms9900>

```

Step 6: Congratulations, you have now successfully built a 32K ROM image. Enjoy!

Revision

Date	Author	Remark
10.01.2010	retroclouds	Initial version created (v0.1)
20.03.2010	retroclouds	Removed PERL and cygwin stuff, we now have an executable file (v0.2)